UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

FORTRAN programs to calculate differential stress concentration factors around two-dimensional circular and elliptical inclusions in an edge-loaded plate

by

David L. Campbell

U.S. Geological Survey, Denver, Colorado 80225

Open-File Report 77- 459

1977

This report is preliminary and has not been edited or reviewed for conformity with U.S. Geological Survey standards and nomenclature.

FORTRAN programs to calculate differential stress concentration factors around two-dimensional circular and elliptical inclusions in an edge-loaded plate

by

David L. Campbell

U.S. Geological Survey, Denver, Colorado 80225

Abstract

FORTRAN IV listings and check-case outputs are given for user-interactive programs CIRCLE and ELLIPSE. The programs suppose either circular or elliptical inclusions, respectively, in an edge-loaded elastic plate, and calculate differential stress concentration factors over a 10 x 11 array of physical points which can be specified by the user. Differential stress concentration factor is defined in this case to be the amount by which the radius of Mohr's circle expands in the vicinity of the inclusion. The two programs handle two-dimensional geometry only. Both programs print extensive prompt-lists to help the user on input.

Equations for stresses in the vicinity of a circular cylindrical elastic inclusion in an edge-loaded elastic plate may be found in Jaeger and Cook[1]. Similarly, the solution for an elliptical, cylindrical inclusion is given by Donnell[2]. In both cases, the generators of the cylinder are taken perpendicular to the plane of the stresses, so that the problems under consideration are strictly two-dimensional. The stresses depend on the contrast in rigidity, XK, between inclusion and plate, and on the Poisson's ratios, PRP and PRI, of plate and inclusion, respectively.

In problems of failure mechanics, it is of interest to know (a) the direction of stress trajectories and (b) the amount by which the radius of Mohr's circle has increased due to the inclusion. Programs CIRCLE and ELLIPSE were written using the equations given by Jaeger and Cook and by Donnell, respectively, to calculate these two quantities. In addition, the Donnell equations were generalized somewhat to allow plate and inclusion to have differing Poisson's ratios, and to allow calculation of stresses at all points when the applied stress is at right angles to the ellipse's semi-major axis.

It is assumed that principal stresses P1 and P3, where P1>P3, are applied on the plate's edge. Tensional stresses are positive. Thus, a uniaxial <u>compression</u> of one unit would have P1 = 0, P3 = -1, and the radius of Mohr's circle everywhere in the plate if the inclusion were absent would be R = (P1-P3)/2 = 0.5. We <u>define</u> the "differential stress concentration factor" at a given point as the radius of Mohr's circle in the presence of the inclusion divided by its radius in its absence, or

$$F = \pm \sqrt{[(\sigma_{\alpha\alpha} - \sigma_{\beta\beta})/2]^2 + \tau_{\alpha\beta}^2}/2$$

where $\sigma_{\alpha\alpha}$ and $\sigma_{\beta\beta}$ are calculated normal stresses and $\tau_{\alpha\beta}$ the calculated shear stress of a unit prism at the point. As a flag to the user, the sign of F is output as negative in case $\sigma_{\alpha\alpha} + \sigma_{\beta\beta}$ has opposite sign from P1 + P3. A switch is provided to normalize factor F by (P1 + P3)/2 rather than by R when the user wishes--for example, to calculate the case P1 = P3; however, then F's simple meaning as Mohr circle amplification factor is lost.

In program CIRCLE, principal stresses P1 and P3 are renamed PX and PY, and are associated with x and y-axes. CIRCLE allows either PX or PY to be the larger. Differential stress concentration factors are calculated over a 10 x 11 grid of circular cylindrical coordinate points (RPERA, THETA) which may be set by the user, but have default values in one quadrant which are probably adequate for most investigations. (Values repeat by symmetry through the other quadrants.)

In program ELLIPSE, the convention P1>P3 should be adhered to. Results are again calculated over an adjustable 10 x 11 grid, whose default values cover one quadrant. Elliptical coordinates (ALPHA, THETA) are used, where

$$x = C \cosh(\mathrm{ALPHA}) \cos(\mathrm{THETA}),$$

$$y = C \sinh(\mathrm{ALPHA}) \sin(\mathrm{THETA}),$$

$$C = \sqrt{1 - E^2},$$

and E is the ellipticity (= semi-minor axis/semi-major axis) of the ellipse. The angle THETA is in degrees. The x-axis is along the long, or semi-major, axis of the ellipse. Default values of ALPHA step in 10 equal increments from the inclusion boundary to a value whose x-intercept corresponds to 3 times the semi-major axis. Input variable ANGLE is the clockwise angle in degrees between the x-axis and the direction of P1. Figure 1 illustrates the resulting geometrical relations for the case ALPHA = $\pm$ 30, when the plate is under uniaxial unit compression. Unless ANGLE = 0° or 90°, a second computation with the sign of ANGLE reversed must be run in order to see the complete stress concentration pattern, since the results of a given run repeat by symmetry in these cases to only one other quadrant (the opposite one), and not to all three.

The programs were written in FORTRAN IV and were run on the U.S. Geological Survey Honeywell 68/80 Multics computer at Denver, Colorado. They were written for user-interactive use, and print out many messages to prompt appropriate user replies. Input is by FORTRAN namelist. Listings of the programs and some check-cases follow. Two non-standard features of Multics FORTRAN seen on the listings are continuation lines without a continuation character, and the use of both upper and lower case letters.

References cited

1. Jaeger, J. C., and Cook, N. G. W., 1969, Fundamentals of Rock Mechanics: Methuen, p. 236-251.

2. Donnell, L. H., 1941, Stress concentrations due to elliptical discontinuities in plates under edge forces: Theodore von Karman Anniversary Volume, Pasadena, p. 293-309.

PROGRAM CIRCLE

```
calculation of stress concentration around a circular inclusion in a plate......
..
c.....equations from Jaeger and Cook, Fundamentals of Rock Mechanics,
c       (Methuen, 1969), Section 10.9, pages 248-251.
      real norm
      integer problm,switch,detail
      automatic rpera,theta
      dimension bigf(11,10), rpera(11),theta(10),traj(11,10),sigmar(11,10),sigmat
(11,10),taurt(11,10)
      namelist/input/xk,prp,pri,px,py,problm,switch,detail
      namelist /grid/theta,rpera
c.....Set default input values.
      problm=1
      switch=1
      px=-1.
      py=0.
      deg2rd=3.14159/180.
      data (rpera(i),i=1,11)/1.0,1.1,1.2,1.3,1.4,1.5,1.6,1.8,2.0,2.5,3.0/
      data (theta(i),i=1,10)/0.,10.,20.,30.,40.,50.,60.,70.,80.,90./
c....Print user prompt messages.
     -print 75
75    format (////" Stress Concentration around a Circular Inclusion."
//" Program by Dave Campbell, US Geol. Survey, Denver.  Feb77"
/" Equations from Jaeger and Cook, Section 10.9,"
//" Variables in namelist input:"
/"       xk = (Rigidity of inclusion)/(Rigidity of plate)"
/"       prp = Poisson's ratio of plate."
/"       pri = Poisson's ratio of inclusion."
/"       px = Force in x-direction on plate's y-edges. (default=-1.)"
/"       py = Force in y-direction on plate's x-edges. (default=0.)."
/"          (Note: negative forces are compressional, positive ones tensional.)")
      print 76
76    format (
"       problm = 1 for plane strain (default),"
/"                 2 for plane stress,"
/"                 0 to stop."
/"       detail = 0 to output stress concentration factor only. (default)"
/"                 1 to output stress trajectory angles, too."
/"                 2 to output stress fields, too."
/"       switch = 1 to normalize by (px-py)/2 (default)"
/"                 2 to normalize by (px+py)/2.")
c.....Subsequent passes enter here.
400   continue
      print 10
10    format(////" type namelist input:xk,prp,pri...(problm=0 to stop.)",/)
      read input
      if(problm.eq.0)stop
      t=(px+py)/2.
      s=(px-py)/2.
      norm=s
      if (switch .eq. 2) norm=t
      yo=3.-4.*prp
      yi=3.-4.*pri
      if (problm .eq. 1) go to 100
      yo=(3.-prp)/(1.+prp)
      yi=(3.-pri)/(1.+pri)
100   continue
      d1=xk*yo+1.
      d2=2.*xk+yi-1.
      a=2.*(1.-xk)/d1
```

```
      b=(yi-1.-xk*(yo-1.))/d2
      c=(xk-1.)/d1
      f=2.*d1*d2
      e=xk*(yo+1.)
      h1=((xk*(yo+2.)+yi    )*e)/f
      h2=((xk*(yo-2.)-yi+2.)*e)/f
c.....Y & C eqn (16).
      sigma1=h1*px+h2*py
c.....Y & C eqn (17).
      sigma2=h2*px+h1*py
      y=(sigma1-sigma2)/2.
      fbig=(sign(y,sigma1+sigma2))/norm
      do 1000 i=1,11
      xi=rpera(i)
      x=1./(xi**2)
      sr=t*(1.-b*x)
      st=t*(1.+b*x)
      tc=3.*c*x*x
      fr=s*(1.-2.*a*x-tc)
      ft=s*(1.-tc)
      tt=s*(1.+a*x+tc)
      do 1000 j=1,10
      tj=theta(j)
      ttt=2.*tj*deg2rd
      cst=cos(ttt)
      snt=sin(ttt)
c.....Y & C eqn (20).
      sigr=sr+fr*cst
c.....Y & C eqn (21).
      sigt=st-ft*cst
c.....Y & C eqn (22).
      tau=-tt*snt
      sigmar(i,j)=sigr
      sigmat(i,j)=sigt
      taurt(i,j)=tau
      y=sqrt(((sigr-sigt)/2.)**2+tau**2)
      bigf(i,j)=(sign(y,sigr+sigt))/(sign(norm,t))
      traj(i,j)=atan2(2.*tau,sigr-sigt)/(2.*deg2rd)+theta(j)
1000  continue
c.....printing of output begins here.
      print 40
40    format(////" DIFFERENTIAL STRESS CONCENTRATION FACTORS AROUND A CIRCULAR I
NCLUSION",/)
      if (problm .eq. 1) print 62
      if (problm .eq. 2) print 64
62    format(1h ,29x," plane strain calculation")
64    format(1h ,29x," plane stress calculation")
      print 45, xk,prp,pri
45    format(1h ,29x,"inclusion rigidity/plate rigidity = ",f8.3,/30x,"Poisson's
 ratio of plate = ",f8.3,/30x,"Poisson's ratio of inclusion = ",f8.3)
      if (switch .eq. 0) print 46
46    format (30x,"normalization factor=(px+py)/2")
      print 30,(theta(k),k=1,10)
30    format(/" theta=",10f7.2)
      print 35
35    format("  r/a !",72(1h_))
      print 25, fbig
25    format (1h ,"<1.0 !",f7.3,"    everywhere......")
      print 20, (rpera(i),(bigf(i,j),j=1,10),i=1,11)
20    format(1h ,f4.1,x,1h!,10f7.3)
```

```
      if (detail .eq. 0) go to 300
      print 39
39  . format (//" stress trajectory: angle in degrees from x direction")
      print 30,(theta(k),k=1,10)
      print 35
      print 21, (rpera(i),(traj(i,j),j=1,10),i=1,11)
21    format(1h ,f5.3,1h!,10f7.2)
      if (detail .eq. 1) go to 300
      print 36
36    format ( //" rpera component of stress:")
      print 30,(theta(k),k=1,10)
      print 35
      print 20, (rpera(i),(sigmar(i,j),j=1,10),i=1,11)
      print 37
37.   format(//" theta component of stress:")
      print 30,(theta(k),k=1,10)
      print 35
      print 20, (rpera(i),(sigmat(i,j),j=1,10),i=1,11)
      print 38
38  — format (//" tau component of stress:")
      print 30,(theta(k),k=1,10)
      print 35
      print 20, (rpera(i),(taurt(i,j),j=1,10),i=1,11)
c.....Now allow user to change gridpoints, in order to investigate details
c     in critical regions.
300   continue
      print 80
80    format (/" grid points may now be changed via namelist grid."
/" variables in namelist grid:"
/"      theta = a 10-length vector of values theta as above."
/"      rpera = an 11-length vector of values r/a as above."
/"          (program gives incorrect answers for rpera points inside inclusion.)"
//" input namelist grid. (type ",8h"$grid$"," for no change.)")
      read grid
      go to 400
      end
```

EXAMPLE AND CHECK CASE: CIRCLE

USER PROMPT MESSAGE

```
Stress Concentration around a Circular Inclusion.

Program by Dave Campbell, US Geol. Survey, Denver.  Feb77
Equations from Jaeger and Cook, Section 10.9,

Variables in namelist input:
     xk = (Rigidity of inclusion)/(Rigidity of plate)
     prp = Poisson's ratio of plate.
     pri = Poisson's ratio of inclusion.
     px = Force in x-direction on plate's y-edges. (default=-1.)
     py = Force in y-direction on plate's x-edges. (default=0.).
        (Note: negative forces are compressional, positive ones tensional.)
     problm = 1 for plane strain (default),
              2 for plane stress,
              0 to stop.
     detail = 0 to output stress concentration factor only. (default)
              1 to output stress trajectory angles, too.
              2 to output stress fields, too.
     switch = 1 to normalize by (px-py)/2 (default)
              2 to normalize by (px+py)/2.

type namelist input:xk,prp,pri...(problm=0 to stop.)
```

INPUT RESPONSE →

```
$input xk=.1,prp=.25,pri=.4,detail=1$
```

OUTPUT

```
DIFFERENTIAL STRESS CONCENTRATION FACTORS AROUND A CIRCULAR INCLUSION

                               plane strain calculation
                              inclusion rigidity/plate rigidity =     0.100
                              Poisson's ratio of plate =    0.250
                              Poisson's ratio of inclusion =    0.400
```

theta=	0.00	10.00	20.00	30.00	40.00	50.00	60.00	70.00	80.00	90.00
r/a !										
<1.0 !	0.250	everywhere......								
1.0 !	-1.250	-1.148	-0.856	0.433	0.315	0.841	1.392	1.848	2.146	2.250
1.1 !	-0.884	-0.841	0.736	0.653	0.717	0.942	1.224	1.478	1.650	1.710
1.2 !	-0.696	0.713	0.763	0.847	0.957	1.080	1.201	1.301	1.367	1.391
1.3 !	0.604	0.666	0.809	0.965	1.092	1.174	1.210	1.213	1.202	1.196
1.4 !	0.565	0.655	0.845	1.033	1.167	1.228	1.219	1.164	1.102	1.075
1.5 !	0.556	0.658	0.870	1.072	1.207	1.256	1.222	1.134	1.041	1.000
1.6 !	0.562	0.669	0.887	1.092	1.225	1.267	1.220	1.113	1.002	0.953
1.8 !	0.597	0.698	0.906	1.104	1.230	1.262	1.204	1.085	0.961	0.906
2.0 !	0.641	0.729	0.918	1.100	1.216	1.243	1.183	1.066	0.944	0.891
2.5 !	0.738	0.798	0.936	1.075	1.166	1.185	1.135	1.038	0.940	0.898
3.0 !	0.806	0.848	0.949	1.055	1.126	1.140	1.100	1.023	0.948	0.917

stress trajectory: angle in degrees from x direction

theta=	0.00	10.00	20.00	30.00	40.00	50.00	60.00	70.00	80.00	90.00
r/a !										
1.000!	90.00	97.86	104.59	105.00	65.73	58.51	64.47	72.49	81.14	90.00
1.100!	90.00	91.69	91.05	85.57	77.40	73.66	74.91	78.90	84.19	90.00
1.200!	90.00	86.34	83.16	80.95	80.00	80.36	81.82	84.10	86.92	90.00
1.300!	90.00	82.82	79.56	79.60	81.33	83.67	85.97	87.83	89.12	90.00
1.400!	90.00	80.98	78.11	79.32	82.23	85.55	88.46	90.32	90.73	90.00
1.500!	90.00	80.29	77.72	79.47	82.93	86.73	90.00	91.92	91.84	90.00
1.600!	90.00	80.27	77.86	79.84	83.51	87.52	90.96	92.92	92.55	90.00
1.800!	90.00	81.13	78.85	80.78	84.44	88.48	91.94	93.86	93.20	90.00
2.000!	90.00	82.31	80.09	81.78	85.19	89.01	92.30	94.06	93.28	90.00
2.500!	90.00	84.78	82.85	83.90	86.52	89.60	92.24	93.55	92.74	90.00
3.000!	90.00	86.33	84.76	85.41	87.39	89.82	91.88	92.83	92.12	90.00

USER PROMPT

```
grid points may now be changed via namelist grid.
variables in namelist grid:
     theta = a 10-length vector of values theta as above.
     rpera = an 11-length vector of values r/a as above.
        (program gives incorrect answers for rpera points inside inclusion.)
input namelist grid. (type "$grid$" for no change.)
```

INPUT RESPONSE →

```
$grid$
```

SECOND PROMPT →

```
type namelist input:xk,prp,pri...(problm=0 to stop.)
```

INPUT RESPONSE TO END PROGRAM →

```
$input problm=0$

STOP
r 1528 0.938 3.434 125
```

PROGRAM ELLIPSE

```
calculation of stress concentration around an elliptical inclusion.....
c     equations from donnell, 1941, von karman volume, pps. 293-309.
      real norm
      integer problm,switch,detail
      automatic alpha,theta
      dimension bigf(11,10),traj(11,10),alpha(11),theta(10),sigmaa(11,10),sigmab
(11,10),tauab(11,10)
      common /geom/xxk,yo,yi,a0
      common /point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      namelist/input/xk,prp,pri,p1,p3,angle,e,problm,switch,detail
      namelist/grid/theta,alpha
      acosh(x)=alog(x+sqrt(x**2-1.))
c.....set default input values.
      data (theta(i),i=1,10)/0.,10.,20.,30.,40.,50.,60.,70.,80.,90./
      problm=1
      switch=1
      detail=0
      p1=0.
      p3=-1.
      angle=90.
      e=.5
      deg2rd=3.14159/180.
      itime=0
c.....print user prompt messages.
      print 75
75    format (////" Stress Concentration around Elliptical Inclusion."
//" Program by Dave Campbell, US Geol. Survey, Denver.  Feb77"
/" Equations from donnell, 1941, von karman volume, pps.293-309."
//" Variables in namelist input:"
/"      xk = (Rigidity of inclusion)/(Rigidity of plate)"
/"      pri = Poisson's ratio of inclusion."
/"      prp = Poisson's ratio of plate."
/"      p1 = maximum principal stress on plate at infinity.  (default=0.)"
/"      p3 = minimum principal stress on plate at infinity.  (default=-1.)"
/"      angle = angle in degrees between p1 and ellipse major axis.  (default=90
)"
/"         (negative stresses are compressional, positive ones tensional."
/"          hence default case has compression along ellipse's major axis.)")
      print 76
76    format(
 "      e = ellipticity of ellipse (=semi-minor axis/semi-major axis)."
/"          (Must be between 0.(thin sawcut) and 1.(circle). default=.5)"
/"      problm = 1 for plane strain (default),"
/"               2 for plane stress,"
/"               0 to stop."
/"      detail = 0 to output stress concentration factor only. (default)"
/"               1 to output stress trajectory angles, too."
/"               2 to output stress fields, too."
/"      switch = 1 to normalize by mohr circle radius (default)"
/"               2 to normalize by hydrostatic stress (p1+p3)/2.")
c.....subsequent passes enter here.
400   continue
      print 10
10    format(////" type namelist input:xk,pri,prp...(problm=0 to stop.)",/)
      read input
      if(problm.eq.0)stop
c.....Break down general case of ellipse at angle to principal stress
c     directions in terms of bndry loads px,py, & tf, for which
c     Donnell gives solutions.
convention: x-axis is taken along ellipse's major axis.
```

```
c      px = Normal Force in x-direction on plate's y-edges.
c      py = Normal Force in y-direction on plate's x-edges.
c         (Note: negative forces are compressional, positive ones tensional.)
c      tf = Tangential (shear) Force on plate's edges.
c         (Note: equal and opposite on adjacent edges so plate won't rotate.)
      ang=angle*deg2rd*2.
      t=(p1+p3)/2.
      s=abs((p1-p3)/2.)
      px=t+s*cos(ang)
      py=t-s*cos(ang)
      tf=t*sin(ang)
      norm=sign(s,t)
      if (switch .eq. 2) norm=t
c.....Donnell takes prp=pri.  The next 7 steps, plus changes noted
c     in subroutines pcons & tcons, set up the general case.
      xxk=xk*(1.+prp)/(1.+pri)
      yo=3.-4.*prp
      yi=3.-4.*pri
      if (problm .eq. 1) go to 100
      yo=(3.-prp)/(1.+prp)
      yi=(3.-pri)/(1.+pri)
100   continue
c.....Set up default values of alpha, running from bndry to x-
c     value 3 times semi-major axis.  Skip on subsequent passes.
      if (itime .eq. 1) go to 200
      a0=acosh(1./sqrt(1.-e**2))
      alpha(1)=a0
      alpha(11)=acosh(3./sqrt(1.-e**2))
      ainc=(alpha(11)-a0)/10.
      do 1002 i=2,10
      alpha(i)=alpha(i-1) + ainc
1002  continue
200   continue
      call pt(a0,0.)
      call pcons(+1,py,y1,y1p,y2,y2p,y3,y3p,y4,y4p,y5,y5p)
      call tcons(tf,t1,t1p,t2,t2p,t3,t3p,t4,t4p,t5,t5p)
      call pcons(-1,px,x1,x1p,x2,x2p,x3,x3p,x4,x4p,x5,x5p)
      call pt(a0/2.,0.)
      call pterm(y1p,y2p,y3p,y4p,y5p,ysiga,ysigb,ytau)
      call tterm(t1p,t2p,t3p,t4p,t5p,tsiga,tsigb,ttau)
      call pterm(x1p,x2p,x3p,x4p,x5p,xsiga,xsigb,xtau)
      siga=xsiga+ysiga+tsiga
      tau=xtau+ytau+ttau
      sigb=xsigb+ysigb+tsigb
      s=siga+sigb
      y=sqrt(((siga-sigb)/2.)**2 + tau**2)
      fbig=sign(y,s)/norm
      do 1000 j=1,10
      tj=theta(j)
      beta=tj*deg2rd
      do 1000 i=1,11
      alfa=alpha(i)
      call pt(alfa,beta)
      call pterm(y1,y2,y3,y4,y5,ysiga,ysigb,ytau)
      call tterm(t1,t2,t3,t4,t5,tsiga,tsigb,ttau)
      call pterm(x1,x2,x3,x4,x5,xsiga,xsigb,xtau)
      if (alfa .ge. a0) go to 500
      call pterm(y1p,y2p,y3p,y4p,y5p,ysiga,ysigb,ytau)
      call tterm(t1p,t2p,t3p,t4p,t5p,tsiga,tsigb,ttau)
      call pterm(x1p,x2p,x3p,x4p,x5p,xsiga,xsigb,xtau)
```

```
500    continue
c.....Do superposition of px,py, & tf solutions.
       siga=xsiga+ysiga+tsiga
       tau=xtau+ytau+ttau
       sigb=xsigb+ysigb+tsigb
       sigmaa(i,j)=siga
       sigmab(i,j)=sigb
       tauab(i,j)=tau
       s=siga+sigb
       y=sqrt(((siga-sigb)/2.)**2 + tau**2)
       bigf(i,j)=sign(y,s)/norm
       traj(i,j)=(atan2(2.*tau,siga-sigb)/2.
       +atan2((exp(alfa)+exp(-alfa))*sin(beta),(exp(alfa)-exp(-alfa))*cos(beta))
       )/deg2rd
1000   continue
c.....output section begins here.
       print 40
40     format(////" DIFFERENTIAL STRESS CONCENTRATION FACTORS AROUND AN ELLIPTICA
L INCLUSION",/)
       if (problm .eq. 1) print 62
       if (problm .eq. 2) print 64
62     format(1h ,29x,"plane strain calculation")
64     format(1h ,29x,"plane stress calculation")
       print 45, xk,prp,pri,e,p1,p3,angle
45     format(1h ,29x,"inclusion rigidity/plate rigidity = ",f8.3,/30x,"Poisson's
 ratio of plate = ",f8.3,/30x,"Poisson's ratio of inclusion = ",f8.3,/30x,"ellip
ticity of ellipse = ",f8.3,/30x,"principal stress p1 = ",f8.3,/30x,"principal st
ress p3 = ",f8.3,/30x,"ellipse tilt angle = ",f4.0," deg")
       if (switch .eq. 0) print 46
46     format (30x,"normalization factor=(p1+p3)/2")
       print 30,(theta(k),k=1,10)
30     format(/" theta=",10f7.2)
       print 35
35     format(" alpha!",72(1h_))
       print 25,fbig
25     format("  <a0 !",f7.3,"     everywhere.......")
       print 20, (alpha(i),(bigf(i,j),j=1,10),i=1,11)
20     format(1h ,f5.3,1h!,10f7.3)
       itime=1
       if (detail .eq. 0) go to 300
       print 39
39     format (//" stress trajectory: angle in degrees from x direction (along ma
jor axis)")
       print 30,(theta(k),k=1,10)
       print 35
       print 21, (alpha(i),(traj(i,j),j=1,10),i=1,11)
21     format(1h ,f5.3,1h!,10f7.2)
       if (detail .eq. 1) go to 300
       print 36
36     format ( //" alpha component of stress:")
       print 30,(theta(k),k=1,10)
       print 35
       print 20, (alpha(i),(sigmaa(i,j),j=1,10),i=1,11)
       print 37
37     format(//" beta component of stress:")
       print 30,(theta(k),k=1,10)
       print 35
       print 20, (alpha(i),(sigmab(i,j),j=1,10),i=1,11)
       print 38
38     format (//" tau component of stress:")
```

```
      print 30,(theta(k),k=1,10)
      print 35
      print 20, (alpha(i),(tauab(i,j),j=1,10),i=1,11)
c.....Now allow user to cnange griopoints so he can look at details
c     of the solution in small regions.
300   continue
      print 80
80    format (/" grid points may now be changed via namelist grid."
/" variables in namelist grid:"
/"        theta = a 10-length vector of values theta as above."
/"        alpha = an 11-length vector of values alpha as above."
//" input namelist grid. (type ",8h"$grid$"," for no change.)")
      read grid
      go to 400
      end
```

```
      subroutine pt(a,b)
      common/point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      e2a=exp(2.*a)
      em2a=exp(-2.*a)
      e4a=exp(4.*a)
      em4a=exp(-4.*a)
      c2b=cos(2.*b)
      s2b=sin(2.*b)
      c4b=cos(4.*b)
      s4b=sin(4.*b)
      return
      end
```

```
      subroutine pterm(ap1,am1,b0,bp2,bm2,siga,sigb,tau)
c.....this subroutine programs donnell's eqns (14).
      common /point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      factr=1./(((e2a+em2a)/2.)-c2b)**2
      siga=factr*(-ap1*((2.+e4a)-2.*(e2a+em2a)*c2b+c4b)
                 +am1*((2.+em4a)-2.*(e2a+em2a)*c2b+c4b) - b0*(e2a-em2a)
                 -bp2*(3.*e2a-(3.+e4a)*c2b+e2a*c4b)
                 +bm2*(3.*em2a-(3.+em4a)*c2b+em2a*c4b))
      sigb=factr*(-ap1*((2.+e4a)-4.*e2a*c2b+c4b)
             +am1*((2.+em4a)-4.*em2a*c2b+c4b) + b0*(e2a-em2a)
                 +bp2*(3.*e2a-(3.+e4a)*c2b+e2a*c4b)
                 -bm2*(3.*em2a-(3.+em4a)*c2b+em2a*c4b))
      tau=-factr*(((ap1+am1)*(e2a+em2a)+2.*b0)*s2b
      +bp2*((3.+e4a)*s2b-e2a*s4b)+bm2*((3.+em4a)*s2b-em2a*s4b))
      return
      end
```

```
      subroutine tterm(ap1,am1,b0,bp2,bm2,siga,sigb,tau)
c.....this subroutine programs donnell's eqns (21).
      common /point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      factr=1./(((e2a+em2a)/2.) - c2b)**2
      siga=factr*(-(ap1+am1)*(2.*(e2a+em2a)*s2b-s4b)-2.*b0*s2b
                 -bp2*((3.+e4a)*s2b-e2a*s4b) - bm2*((3.+em4a)*s2b-em2a*s4b))
      sigb=factr*((ap1+am1)*s4b+2.*b0*s2b+bp2*((3.+e4a)*s2b-e2a*s4b)
                 +bm2*((3.+em4a)*s2b-em2a*s4b))
      tau=factr*(((ap1+am1)*c2b+b0)*(e2a-em2a)+bp2*(3.*e2a-(3.+e4a)*c2b
                 +e2a*c4b)-bm2*(3.*em2a-(3.+em4a)*c2b+em2a*c4b))
      return
      end
```

```
      subroutine pcons(n,py,ap1,ap1pr,am1,am1pr,b0,b0pr,bp2,bp2pr,bm2,bm2pr)
c.....This subroutine programs Donnell's eqns (18), generalized to in-
clude additional factor n.  His eqns hold only if load is in y dire-
ction, and n=1.  When load is in x direction, signs change, and n=-1.
c.....A second generalization, to include unmatched Poisson's ratios,
changes yo to yi in the denominator of the ap1pr expression.
      common /geom/xk,yo,yi,a0
      common /point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      ap1=-py/8.
      bp2=ap1*n
      if (xk .eq. 0) go to 100
      bigm=(yi-1.-xk*(yo-1.))/(xk*(yo+1.))
      ap1pr=((xk*yo+1.)*e2a+(1.-xk)*(2.*n+em2a))*ap1
     /((yi+1.)*em2a+(xk*yo+1.)*(bigm+1.)*(e2a-em2a))
      am1=-e4a*ap1-(bigm+1.)*(1.-e4a)*ap1pr
      b0=-bigm*(e2a-em2a)*ap1pr
200   continue
      am1pr=-ap1pr
      bp2pr=n*ap1+em2a*(ap1-am1-2.*ap1pr)/2.
      bm2pr=-bp2pr
      b0pr=0.
      bm2=-n*e4a*ap1-(1.-e4a)*bp2pr
      return
100   continue
      ap1pr=0.
      am1=-ap1*(em2a+(1.-e4a)*(e2a+em2a+2.)/(e2a-em2a))
      b0=-ap1*(e2a+em2a+2.)
      go to 200
      end
```

```
      subroutine tcons(tf,ap1,ap1pr,am1,am1pr,b0,b0pr,bp2,bp2pr,bm2,bm2pr)
c.....This subroutine programs Donnell's eqns (24), generalized to in-
clude cases when Poisson's ratios of plate and inclusion don't match.
c.....Only eqns for am1pr & ap1pr are effected by this generalization.
      common/geom/xk,yo,yi,a0
      common /point/e2a,em2a,e4a,em4a,c2b,s2b,c4b,s4b
      f=(1.+yo)/(1.+yi)
      ap1=0.
      b0=0.
      b0pr=0.
      bp2=-tf/4.
      bigp=-bp2/((1.-xk)*em4a-(xk*yo+1.))
      am1pr=2.*f*xk*(1.-xk)*bigp*em2a
      ap1pr=-am1pr
      am1=-2.*(1.-xk)*bigp*(e2a-em2a)
      bp2pr=xk*(yo+1.)*bigp
      bm2pr=-bp2pr
      bm2=bigp*((1.-xk)*e4a-xk*yo-1.)
      return
      end
```

EXAMPLE AND CHECK CASE: ELLIPSE

USER PROMPT MESSAGE

```
Stress Concentration around Elliptical Inclusion.

Program by Dave Campbell, US Geol. Survey, Denver.  Feb77
Equations from donnell, 1941, von karman volume, pps.293-309.

Variables in namelist input:
     xk = (Rigidity of inclusion)/(Rigidity of plate)
     pri = Poisson's ratio of inclusion.
     prp = Poisson's ratio of plate.
     p1 = maximum principal stress on plate at infinity.  (default=0.)
     p3 = minimum principal stress on plate at infinity.  (default=-1.)
     angle = angle in degrees between p1 and ellipse major axis.  (default=90)
       (negative stresses are compressional, positive ones tensional.
        hence default case has compression along ellipse's major axis.)
     e = ellipticity of ellipse (=semi-minor axis/semi-major axis).
        (Must be between 0.(thin sawcut) and 1.(circle). default=.5)
     problm = 1 for plane strain (default),
              2 for plane stress,
              0 to stop.
     datail = 0 to output stress concentration factor only. (default)
              1 to output stress trajectory angles, too.
              2 to output stress fields, too.
     switch = 1 to normalize by mohr circle radius (default)
              2 to normalize by hydrostatic stress (p1+p3)/2.

type namelist input:xk,pri,prp...(problm=0 to stop.)
```

INPUT RESPONSE →

```
$input xk=.1,prp=.25,pri=.4,detail=1$
```

OUTPUT

```
DIFFERENTIAL STRESS CONCENTRATION FACTORS AROUND AN ELLIPTICAL INCLUSION

                               plane strain calculation
                               inclusion rigidity/plate rigidity =    0.100
                               Poisson's ratio of plate =    0.250
                               Poisson's ratio of inclusion =    0.400
                               ellipticity of ellipse =    0.500
                               principal stress p1 =    0.000
                               principal stress p3 =   -1.000
                               ellipse tilt angle =  90. deg
```

theta= / alpha!	0.00	10.00	20.00	30.00	40.00	50.00	60.00	70.00	80.00	90.00
<a0 !	0.188	everywhere.......								
0.549!	-1.156	-0.841	0.228	0.550	1.019	1.341	1.551	1.681	1.751	1.773
0.686!	-0.528	-0.634	0.763	0.860	0.987	1.135	1.267	1.364	1.423	1.442
0.822!	0.441	0.667	0.939	1.064	1.102	1.123	1.152	1.185	1.210	1.219
0.959!	0.497	0.697	0.977	1.131	1.168	1.148	1.116	1.093	1.081	1.078
1.095!	0.585	0.733	0.977	1.135	1.185	1.160	1.105	1.049	1.011	0.997
1.232!	0.669	0.775	0.971	1.117	1.173	1.155	1.096	1.027	0.975	0.956
1.368!	0.741	0.816	0.968	1.093	1.150	1.139	1.085	1.016	0.961	0.940
1.505!	0.800	0.853	0.968	1.072	1.124	1.119	1.073	1.010	0.957	0.937
1.641!	0.846	0.884	0.971	1.054	1.100	1.099	1.061	1.006	0.960	0.941
1.778!	0.882	0.910	0.975	1.041	1.079	1.080	1.049	1.004	0.964	0.949
1.914!	0.909	0.930	0.979	1.031	1.062	1.064	1.040	1.003	0.970	0.957

```
stress trajectory: angle in degrees from x direction (along major axis)
```

theta= / alpha!	0.00	10.00	20.00	30.00	40.00	50.00	60.00	70.00	80.00	90.00
0.549!	-90.00	105.40	100.25	58.98	63.87	70.11	75.75	80.81	85.50	90.00
0.686!	-90.00	81.30	79.63	78.55	77.94	78.89	81.05	83.83	86.87	90.00
0.822!	-90.00	76.04	77.76	81.45	83.84	85.08	86.00	87.10	88.47	90.00
0.959!	-90.00	77.39	78.33	82.39	85.94	88.17	89.28	89.72	89.89	90.00
1.095!	-90.00	80.10	79.83	83.20	86.82	89.53	91.01	91.36	90.89	90.00
1.232!	-90.00	82.59	81.60	84.09	87.34	90.08	91.74	92.15	91.42	90.00
1.368!	-90.00	84.54	83.29	85.02	87.74	90.27	91.93	92.38	91.59	90.00
1.505!	-90.00	85.98	84.74	85.90	88.10	90.30	91.83	92.27	91.54	90.00
1.641!	-90.00	87.03	85.93	86.69	88.43	90.28	91.61	92.02	91.37	90.00
1.778!	-90.00	87.79	86.87	87.37	88.72	90.24	91.36	91.71	91.17	90.00
1.914!	-90.00	88.35	87.60	87.93	88.97	90.19	91.12	91.41	90.96	90.00

USER PROMPT

```
grid points may now be changed via namelist grid.
variables in namelist grid:
     theta = a 10-length vector of values theta as above.
     alpha = an 11-length vector of values alpha as above.

input namelist grid. (type "$grid$" for no change.)
```

INPUT RESPONSE →

```
$grid$
```

SECOND PROMPT →

```
type namelist input:xk,pri,prp...(problm=0 to stop.)
```

RESPONSE TO END PROGRAM →

```
$input problm=0$

STOP
```



Figure 1 -- Geometry of ellipse axes (x, y) and directions of principal stress P1 and P3. The case shown has P1 = 0 and P3 = -1 (uniaxial unit compression). ANGLE is the clockwise angle in degrees from the x-axis (ellipse major axis) to P1 direction. (a) The case ANGLE = 30. (b) The case ANGLE = -30. In case (b) the direction of x has been reversed in order to show that both cases must be run to produce complete results which will cover the entire plane.